# $Id: PKGBUILD 78767 2012-10-24 11:47:54Z foutrelis $
# Maintainer: Evangelos Foutras <evangelos@foutrelis.com

pkgname=('python-virtualenv-clone' 'python2-virtualenv-clone')
pkgver=0.2.4
pkgrel=3
pkgdesc="A script for cloning a non-relocatable virtualenv."
arch=('any')
url="https://github.com/edwardgeorge/virtualenv-clone"
license=('GPL')
makedepends=('python-distribute' 'python2-distribute')
options=('!emptydirs')
source=(http://pypi.python.org/packages/source/v/virtualenv-clone/virtualenv-clone-$pkgver.tar.gz)
sha256sums=('6797e96bc2a1ab41995bcabdfe08ac1b02c55c719c62036ccf53a0ed5430b935')

package_python-virtualenv-clone() {
  depends=('python-distribute')

  cd "$srcdir/virtualenv-clone-$pkgver"
  python3 setup.py install --root="$pkgdir" -O1
}

package_python2-virtualenv-clone() {
  depends=('python2')

  cd "$srcdir/virtualenv-clone-$pkgver"
  python2 setup.py install --root="$pkgdir" -O1

  # virtualenv-clone is provided by the python-virtualenv-clone package
  rm "$pkgdir/usr/bin/virtualenv-clone"
}

# vim:set ts=2 sw=2 et:
